
    How-To: Complete Graphviz DOT Guide

Mix.install([
  {:yog_ex, "~> 0.98"},
  {:kino_vizjs, "~> 0.8.0"}
])
Introduction
Yog.Render.DOT and Yog.Multi.DOT export graphs to the standard Graphviz DOT language for visualization. By combining the kino_vizjs package with Kino.VizJS, you can render stunning interactive graph diagrams directly in your Livebook.
This guide covers everything from simple rendering to advanced styling, custom subgraphs, algorithm-specific helpers, and parallel-edge multigraph visualizations.

Quick Start
Creating a graph and rendering it with the default settings:
g = Yog.directed()
  |> Yog.add_node("Start", nil)
  |> Yog.add_node("Process", nil)
  |> Yog.add_node("End", nil)
  |> Yog.add_edges!([
    {"Start", "Process", 1},
    {"Process", "End", 1}
  ])

# Render with defaults
dot = Yog.Render.DOT.to_dot(g)
IO.puts(dot)

Kino.VizJS.render(dot)

Default Options & Helpers
Yog.Render.DOT provides a rich configuration map for customizing the generated output.
default_options/0
opts = Yog.Render.DOT.default_options()
IO.inspect(opts.rankdir, label: "Direction")
IO.inspect(opts.node_shape, label: "Shape")
IO.inspect(opts.node_color, label: "Node Color")
default_options_with_edge_formatter/1
Use this when edge data is not a string (e.g. integers, custom structs) to clean up edge labels.
g = Yog.from_edges(:undirected, [{:a, :b, 42}, {:b, :c, 99}])

opts = Yog.Render.DOT.default_options_with_edge_formatter(fn weight ->
  "#{weight} ms"
end)

dot = Yog.Render.DOT.to_dot(g, opts)
Kino.VizJS.render(dot)
default_options_with/2
Customize both node and edge labels at once.
g = Yog.directed()
  |> Yog.add_node(1, %{name: "Alice", role: "Admin"})
  |> Yog.add_node(2, %{name: "Bob", role: "User"})
  |> Yog.add_edge_ensure(1, 2, 100)

opts = Yog.Render.DOT.default_options_with(
  node_label: fn _id, data -> "#{data.name} (#{data.role})" end,
  edge_label: fn weight -> "#{weight} Mbps" end
)

dot = Yog.Render.DOT.to_dot(g, opts)
Kino.VizJS.render(dot)
default_options_without_labels/0
Hide edge labels entirely for a cleaner structure.
g = Yog.from_edges(:directed, [{:a, :b, 5}, {:b, :c, 10}])

opts = Yog.Render.DOT.default_options_without_labels()
dot = Yog.Render.DOT.to_dot(g, opts)
Kino.VizJS.render(dot)

Themes
Professionally designed color schemes built directly into Yog.Render.DOT.
g = Yog.Generator.Classic.binary_tree(3)
Default Theme
dot = Yog.Render.DOT.to_dot(g, Yog.Render.DOT.theme(:default))
Kino.VizJS.render(dot)
Dark Theme
dot = Yog.Render.DOT.to_dot(g, Yog.Render.DOT.theme(:dark))
Kino.VizJS.render(dot)
Minimal Theme
dot = Yog.Render.DOT.to_dot(g, Yog.Render.DOT.theme(:minimal))
Kino.VizJS.render(dot)
Presentation Theme
dot = Yog.Render.DOT.to_dot(g, Yog.Render.DOT.theme(:presentation))
Kino.VizJS.render(dot)

Directions (rankdir)
Control the layout direction of the graph.
g = Yog.from_edges(:directed, [{:a, :b, 1}, {:b, :c, 1}])
Top-Down (:tb)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | rankdir: :tb})
)
Left-to-Right (:lr)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | rankdir: :lr})
)
Bottom-to-Top (:bt)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | rankdir: :bt})
)
Right-to-Left (:rl)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | rankdir: :rl})
)

Node Shapes
Graphviz supports a massive variety of node shapes. You can set them globally or per-node.
g = Yog.directed()
  |> Yog.add_node(1, "Start")
  |> Yog.add_node(2, "Process")
  |> Yog.add_node(3, "Decision")
  |> Yog.add_node(4, "Database")
  |> Yog.add_edges!([{1, 2, 1}, {2, 3, 1}, {3, 4, 1}])
Box (Rectangle)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | node_shape: :box})
)
Circle
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | node_shape: :circle})
)
Cylinder (Database)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | node_shape: :cylinder})
)
Diamond (Decision)
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | node_shape: :diamond})
)
Hexagon
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | node_shape: :hexagon})
)
Parallelogram
Kino.VizJS.render(
  Yog.Render.DOT.to_dot(g, %{Yog.Render.DOT.default_options() | node_shape: :parallelogram})
)

Per-Element Styling
Per-Node Styling (node_attributes)
Pass a callback function fn id, data -> [{attribute, value}] end to customize individual node visual characteristics.
g = Yog.directed()
  |> Yog.add_node(1, %{name: "Alice", role: "Admin"})
  |> Yog.add_node(2, %{name: "Bob", role: "User"})
  |> Yog.add_node(3, %{name: "Carol", role: "Guest"})
  |> Yog.add_edges!([{1, 2, 1}, {2, 3, 1}])

node_attrs = fn _id, data ->
  case data.role do
    "Admin" -> [{:fillcolor, "#ef4444"}, {:style, "filled"}, {:fontcolor, "white"}]
    "User" -> [{:fillcolor, "#3b82f6"}, {:style, "filled"}, {:fontcolor, "white"}]
    _ -> [{:fillcolor, "#94a3b8"}, {:style, "filled"}, {:fontcolor, "black"}]
  end
end

opts = %{
  Yog.Render.DOT.default_options()
  | node_attributes: node_attrs
}

Kino.VizJS.render(Yog.Render.DOT.to_dot(g, opts))
Per-Edge Styling (edge_attributes)
Pass a callback function fn from, to, data -> [{attribute, value}] end for simple graphs to style individual relationships.
g = Yog.from_edges(:directed, [
    {:a, :b, %{priority: :high}},
    {:b, :c, %{priority: :low}},
    {:a, :c, %{priority: :medium}}
  ])

edge_attrs = fn _from, _to, weight ->
  case weight.priority do
    :high -> [{:color, "#ef4444"}, {:penwidth, 3}]
    :medium -> [{:color, "#f59e0b"}, {:penwidth, 2}]
    :low -> [{:color, "#94a3b8"}, {:style, "dashed"}]
  end
end

opts = %{
  Yog.Render.DOT.default_options()
  | edge_attributes: edge_attrs
}

Kino.VizJS.render(Yog.Render.DOT.to_dot(g, opts))

Subgraphs & Clusters
In Graphviz, grouping nodes into subgraphs whose names begin with cluster_ places them in bounding boxes.
g = Yog.directed()
  |> Yog.add_node(:api, "API Gateway")
  |> Yog.add_node(:auth, "Auth Service")
  |> Yog.add_node(:db, "Database")
  |> Yog.add_node(:cache, "Cache")
  |> Yog.add_edges!([{:api, :auth, 1}, {:auth, :db, 1}, {:api, :cache, 1}])

opts = %{
  Yog.Render.DOT.default_options()
  | subgraphs: [
      %{
        name: "cluster_backend",
        label: "Secure Backend Services",
        node_ids: [:auth, :db],
        style: :filled,
        fillcolor: "#e0f2fe",
        color: "#0284c7"
      },
      %{
        name: "cluster_infra",
        label: "Infrastructure",
        node_ids: [:cache],
        style: :dotted,
        fillcolor: "#f1f5f9",
        color: "#475569"
      }
    ]
}

Kino.VizJS.render(Yog.Render.DOT.to_dot(g, opts))

Capstone: System Architecture Diagram
Combining shapes, custom inline styles, edge latency annotations, and clusters:
system = Yog.directed()
  # Users & entry points
  |> Yog.add_node(:users, "Mobile / Web")
  |> Yog.add_node(:cdn, "CDN")
  |> Yog.add_node(:lb, "Load Balancer")
  |> Yog.add_node(:api, "API Gateway")

  # Domain services
  |> Yog.add_node(:auth, "Auth Service")
  |> Yog.add_node(:orders, "Order Service")
  |> Yog.add_node(:payments, "Payment Service")
  |> Yog.add_node(:inventory, "Inventory Service")
  |> Yog.add_node(:notifications, "Notification Service")

  # Data layer
  |> Yog.add_node(:users_db, "Users DB")
  |> Yog.add_node(:orders_db, "Orders DB")
  |> Yog.add_node(:inventory_db, "Inventory DB")
  |> Yog.add_node(:cache, "Redis Cache")
  |> Yog.add_node(:queue, "Event Queue")

  # Connections with protocol metadata
  |> Yog.add_edge_ensure(:users, :cdn, %{protocol: :https, latency: 20})
  |> Yog.add_edge_ensure(:cdn, :lb, %{protocol: :https, latency: 10})
  |> Yog.add_edge_ensure(:lb, :api, %{protocol: :https, latency: 5})
  |> Yog.add_edge_ensure(:api, :auth, %{protocol: :grpc, latency: 15})
  |> Yog.add_edge_ensure(:api, :orders, %{protocol: :grpc, latency: 15})
  |> Yog.add_edge_ensure(:api, :payments, %{protocol: :grpc, latency: 15})
  |> Yog.add_edge_ensure(:orders, :inventory, %{protocol: :grpc, latency: 10})
  |> Yog.add_edge_ensure(:orders, :payments, %{protocol: :grpc, latency: 10})
  |> Yog.add_edge_ensure(:payments, :queue, %{protocol: :amqp, latency: 5})
  |> Yog.add_edge_ensure(:queue, :notifications, %{protocol: :amqp, latency: 5})
  |> Yog.add_edge_ensure(:auth, :users_db, %{protocol: :sql, latency: 5})
  |> Yog.add_edge_ensure(:auth, :cache, %{protocol: :redis, latency: 2})
  |> Yog.add_edge_ensure(:orders, :orders_db, %{protocol: :sql, latency: 5})
  |> Yog.add_edge_ensure(:inventory, :inventory_db, %{protocol: :sql, latency: 5})

# Node shapes based on role
shape_fn = fn id ->
  case id do
    :users -> :circle
    :cdn -> :hexagon
    :lb -> :diamond
    :api -> :box
    n when n in [:auth, :orders, :payments, :inventory, :notifications] -> :box
    n when n in [:users_db, :orders_db, :inventory_db] -> :cylinder
    :cache -> :ellipse
    :queue -> :box
    _ -> :box
  end
end

# Per-edge colors by protocol
edge_attrs = fn _from, _to, weight ->
  case weight.protocol do
    :https -> [{:color, "#3b82f6"}]
    :grpc -> [{:color, "#10b981"}]
    :sql -> [{:color, "#f59e0b"}]
    :redis -> [{:color, "#ef4444"}]
    :amqp -> [{:color, "#8b5cf6"}]
    _ -> []
  end
end

opts = %{
  Yog.Render.DOT.default_options()
  | rankdir: :lr,
    bgcolor: "#f8fafc",
    node_attributes: fn id, _data ->
      shape = shape_fn.(id)
      # Base node attributes
      [{:shape, shape}, {:style, "filled"}, {:fillcolor, "#ffffff"}]
    end,
    edge_attributes: edge_attrs,
    edge_label: fn weight -> "#{weight.latency} ms" end,
    subgraphs: [
      %{name: "cluster_entry", label: "Edge Layer", node_ids: [:users, :cdn, :lb], style: nil, fillcolor: nil, color: nil},
      %{name: "cluster_services", label: "Services", node_ids: [:api, :auth, :orders, :payments, :inventory, :notifications], style: nil, fillcolor: nil, color: nil},
      %{name: "cluster_data", label: "Data Layer", node_ids: [:users_db, :orders_db, :inventory_db, :cache, :queue], style: nil, fillcolor: nil, color: nil}
    ]
}

Kino.VizJS.render(Yog.Render.DOT.to_dot(system, opts))

Highlighting
Manual Highlighting
g = Yog.from_edges(:directed, [{:a, :b, 1}, {:b, :c, 1}, {:c, :d, 1}])

opts = %{
  Yog.Render.DOT.default_options()
  | highlighted_nodes: [:a, :b, :c],
    highlighted_edges: [{:a, :b}, {:b, :c}]
}

Kino.VizJS.render(Yog.Render.DOT.to_dot(g, opts))
path_to_options/2 — Shortest Path Highlighting
g = Yog.Generator.Classic.grid_2d(5, 5)
source = 0
target = 24

{:ok, path} = Yog.Pathfinding.shortest_path(in: g, from: source, to: target)

opts = Yog.Render.DOT.path_to_options(path)
Kino.VizJS.render(Yog.Render.DOT.to_dot(g, opts), height: "600px")

Algorithm Helper Options
mst_to_options/2 — Minimum Spanning Tree
weighted = Yog.from_edges(:undirected, [
  {:a, :b, 4}, {:a, :h, 8}, {:b, :c, 8},
  {:c, :d, 7}, {:c, :f, 4}, {:d, :e, 9},
  {:e, :f, 10}, {:f, :g, 2}, {:g, :h, 1}
])

{:ok, mst} = Yog.MST.kruskal(in: weighted)
mst_opts = Yog.Render.DOT.mst_to_options(mst)
Kino.VizJS.render(Yog.Render.DOT.to_dot(weighted, mst_opts), height: "500px")
community_to_options/2 — Community Detection
sbm = Yog.Generator.Random.sbm(16, 2, 0.8, 0.1, community_sizes: [8, 8])
comm = Yog.Community.Louvain.detect(sbm)
comm_opts = Yog.Render.DOT.community_to_options(comm)
Kino.VizJS.render(Yog.Render.DOT.to_dot(sbm, comm_opts), height: "700px")
cut_to_options/2 — Min-Cut Partitions
flow = Yog.from_edges(:directed, [{:s, :a, 10}, {:s, :b, 10}, {:a, :t, 10}, {:b, :t, 5}])

result = Yog.Flow.MaxFlow.dinic(flow, :s, :t)
min_cut = Yog.Flow.MaxFlow.min_cut(result)
cut_opts = Yog.Render.DOT.cut_to_options(min_cut)
Kino.VizJS.render(Yog.Render.DOT.to_dot(flow, cut_opts))
matching_to_options/2 — Bipartite Matching
bipartite = Yog.from_edges(:undirected, [
  {:a1, :b1, 1}, {:a1, :b2, 1}, {:a2, :b2, 1}, {:a2, :b3, 1}
])

matching = Yog.Matching.hopcroft_karp(bipartite)
match_opts = Yog.Render.DOT.matching_to_options(matching)
Kino.VizJS.render(Yog.Render.DOT.to_dot(bipartite, match_opts))

Multigraph DOT
Yog.Multi.DOT mirrors Yog.Render.DOT but supports parallel edges.
Basic Multigraph Rendering
alias Yog.Multi

mg = Multi.undirected()
  |> Multi.add_node(:london, nil)
  |> Multi.add_node(:paris, nil)

{mg, _e1} = Multi.add_edge(mg, :london, :paris, 100)  # Flight
{mg, _e2} = Multi.add_edge(mg, :london, :paris, 50)   # Train
{mg, e3} = Multi.add_edge(mg, :london, :paris, 300)   # Ferry

Kino.VizJS.render(Yog.Multi.DOT.to_dot(mg))
Per-Edge Styling with edge_id
In multigraphs, the edge_attributes callback receives the unique edge_id as the third parameter and the weight as the fourth: fn from, to, edge_id, weight -> ... end.
opts = %{
  Yog.Multi.DOT.default_options()
  | edge_attributes: fn _from, _to, edge_id, weight ->
      color =
        cond do
          weight == 50 -> "#10b981"   # Train (fastest)
          weight == 100 -> "#3b82f6"  # Flight
          true -> "#f59e0b"           # Ferry
        end

      width = if edge_id == e3, do: "4.0", else: "1.5"
      [{:color, color}, {:penwidth, width}]
    end
}

Kino.VizJS.render(Yog.Multi.DOT.to_dot(mg, opts))
Subgraphs in Multigraphs
mg = Multi.directed()
  |> Multi.add_node(:web, "Web Layer")
  |> Multi.add_node(:api, "API Layer")
  |> Multi.add_node(:db1, "Primary DB")
  |> Multi.add_node(:db2, "Replica DB")
  |> (fn g -> {g, _} = Multi.add_edge(g, :web, :api, 1); g end).()
  |> (fn g -> {g, _} = Multi.add_edge(g, :api, :db1, 1); g end).()
  |> (fn g -> {g, _} = Multi.add_edge(g, :api, :db2, 1); g end).()

opts = %{
  Yog.Multi.DOT.default_options()
  | subgraphs: [
      %{
        name: "cluster_data_layer",
        label: "Data Layer",
        node_ids: [:db1, :db2],
        style: nil,
        fillcolor: nil,
        color: nil
      }
    ]
}

Kino.VizJS.render(Yog.Multi.DOT.to_dot(mg, opts))

Summary
	Capability	Module	Key Function / Option
	Basic export	Yog.Render.DOT	to_dot/2
	Basic export (parallel edges)	Yog.Multi.DOT	to_dot/2
	Default options	Both	default_options/0
	Custom edge formatter	Yog.Render.DOT	default_options_with_edge_formatter/1
	Custom labels	Yog.Render.DOT	default_options_with/2
	Themes	Yog.Render.DOT	theme/1 (:default, :dark, :minimal, :presentation)
	Direction	Both	rankdir: (:tb, :lr, :bt, :rl)
	Node shapes	Both	node_shape: (e.g., :circle, :box, :cylinder, :diamond)
	Per-node styles	Both	node_attributes: callback
	Per-edge styles	Yog.Render.DOT	edge_attributes: fn f, t, w -> ...
	Per-edge styles (multi)	Yog.Multi.DOT	edge_attributes: fn f, t, id, w -> ...
	Subgraphs	Both	subgraphs: list
	Path highlighting	Yog.Render.DOT	path_to_options/2
	MST highlighting	Yog.Render.DOT	mst_to_options/2
	Community colors	Yog.Render.DOT	community_to_options/2
	Min-cut colors	Yog.Render.DOT	cut_to_options/2
	Matching highlight	Yog.Render.DOT	matching_to_options/2



  

    YogEx Cheatsheet

Data Structure
Internal Representation
The core graph is a dual-map adjacency structure designed for efficiency.
	A graph can be :directed or :undirected.
	A Node ID can be any term.
	Node and Edge can hold any term() as data.

# Internal structure visualization (Nodes: 1-6)
%Yog.Graph{
  kind: :directed,
  nodes: %{1 => "A", 2 => "B", 3 => "C", 4 => "D", 5 => "E", 6 => "F"},
  out_edges: %{
    1 => %{2 => 10, 3 => 5},
    2 => %{4 => 8, 6 => 15},
    3 => %{4 => 7, 5 => 12},
    4 => %{5 => 10, 6 => 9},
    5 => %{1 => 14},
    6 => %{2 => 11}
  },
  in_edges: %{
    2 => %{1 => 10, 6 => 11},
    3 => %{1 => 5},
    4 => %{2 => 8, 3 => 7},
    5 => %{3 => 12, 4 => 10},
    6 => %{2 => 15, 4 => 9},
    1 => %{5 => 14}
  }
}
Representation
digraph G {
  rankdir=LR;
  bgcolor="transparent";
  node [shape=circle, fontname="inherit"];
  edge [fontname="inherit", fontsize=10];
  1 [label="A"]; 2 [label="B"]; 3 [label="C"];
  4 [label="D"]; 5 [label="E"]; 6 [label="F"];
  1 -> 2 [label="10"]; 1 -> 3 [label="5"];
  2 -> 4 [label="8"]; 2 -> 6 [label="15"];
  3 -> 4 [label="7"]; 3 -> 5 [label="12"];
  4 -> 5 [label="10"]; 4 -> 6 [label="9"];
  5 -> 1 [label="14"];
  6 -> 2 [label="11"];
}
Performance Profile
	Storage: O(V + E) memory.
	Node Access: O(1) lookup in nodes map.
	Edge Check: O(1) lookup in out_edges.
	Successors: O(1) lookup of neighbor map in out_edges.
	Predecessors: O(1) lookup of parent map in in_edges.
	Transpose: O(1) swapping of in_edges and out_edges

Core Graph Operations
Creation
We can create a graph directly or from edges. 
Yog module contains helper functions to create graphs from edges and has 
:directed or :undirected as first arg.
	weighted - Weight is provided
	unweighted - Weight is not provided (nil)
	simple - Weight is one

# Create directed graph
Yog.directed()

# Create undirected graph
Yog.undirected()

# Custom graph type
Yog.new(:directed)

# From edges
Yog.from_edges(:directed, [{1, 2, 10}, {2, 3, 20}])
Yog.from_unweighted_edges(:undirected, [{1, 2}, {2, 3}])
Yog.from_simple_edges(:undirected, [{1, 2}, {2, 3}])

	Refer to Yog and Yog.Model for more information.
	See Yog.Builder modules for more ways to create graphs.
	See Yog.IO for more ways to import/export graphs.
	See Yog.Generator for more ways to generate graphs.

Modification
A node ID can be any term but it is recommended to use numbers and keep the information
as node data.
add_edge and add_edges return {:ok, graph} or {:error, reason}. Use add_edge! and add_edges! to 
raise an error if the edge cannot be added.
Please note that all modification functions return a new graph.
# Add node with data
graph = Yog.add_node(graph, 1, %{name: "Center", lat: 0, lng: 0})

# Add edge - nodes must exist
{:ok, graph} = Yog.add_edge(graph, from: 1, to: 2, with: %{distance: 10, toll: false, speed_limit: 100})
{:ok, graph} = Yog.add_edges(graph, [{1, 2, 5}, {2, 3, 10}])

## Add edges - nodes are created if they don't exist
{:ok, graph} = Yog.add_edge_ensure(graph, from: 1, to: 0, with: 10, default: 0) # `nil` if default not provided
{:ok, graph} = Yog.add_edge_with(graph, 1, 0, 10, fn node_id -> DataStore.get(node_id) end)

# Remove node/edge
graph = Yog.remove_node(graph, 1)
graph = Yog.remove_edge(graph, 1, 2)
	See Yog.Model for more ways to add nodes and edges.

Transform & Operations
Mapping & Filtering
Modify data or prune structure using functional primitives.
# Transform node data
Yog.Transform.map_nodes(graph, &String.upcase/1)

# Transform edge weights
Yog.Transform.map_edges(graph, fn w -> w * 2 end)

# Prune structure
Yog.Transform.filter_nodes(graph, fn data -> data.active? end)
Yog.Transform.filter_edges(graph, fn _u, _v, w -> w > 5 end)
Graph Transformations
Topological mutations and re-orientations.
# Reverse edge directions (O(1))
Yog.Transform.transpose(graph)

# Manage self-loops
Yog.Transform.remove_self_loops(graph)
Yog.Transform.add_self_loops(graph, 1)

# Convert directionality
Yog.Transform.to_undirected(graph, &min/2)
Yog.Transform.to_directed(graph)
Set Operations
Combine or compare graphs as sets of nodes and edges.
# Operations from Yog.Operation
Yog.Operation.union(graph_a, graph_b)
Yog.Operation.intersection(graph_a, graph_b)
Yog.Operation.difference(graph_a, graph_b)
	See Yog.Operation for composition and isomorphic?.

Pathfinding & Traversal
Pathfinding Options
Algorithms use semiring parameters for custom optimizations.
# Dijkstra/A* options
opts = [
  from: 1, to: 10,
  zero: 0,                # Identity
  add: &+/2,              # Accumulator
  compare: &Yog.Utils.compare/2 # Relation
]
Yog.Pathfinding.shortest_path(graph, opts)
Algorithms & Traversal
Standard search and exploration functions.
# Standard Dijkstra / A*
Yog.Pathfinding.Dijkstra.shortest_path(graph, from: 1, to: 10)
Yog.Pathfinding.BFS.shortest_path(graph, from: 1, to: 10)

# Fold over structure (BFS/DFS)
Yog.Traversal.fold(graph, from: 1, initial: [], with: callback)
Implicit Search & Traversal
Explore dynamic or infinite state spaces defined by logic.
# Expansion-based search (A*)
Yog.Pathfinding.AStar.implicit_a_star(
  from: start_state,
  successors: fn s -> [{next, cost}] end,
  is_goal: fn s -> s == target end
)

# Walk over a virtual graph
Yog.Traversal.implicit_fold(
  from: 1, using: :breadth_first,
  successors_of: fn n -> [n + 1, n * 2] end,
  with: fn count, _id, _meta -> {:continue, count + 1} end
)
Path Results & Hydration
Results return node IDs; "hydration" retrieves the associated data.
# Result internal structure
%{nodes: [1, 5, 10], total_weight: 42.0, distance: 2}

# Map IDs to their associated data
path = Yog.Pathfinding.Dijkstra.shortest_path(graph, from: 1, to: 10)
Enum.map(path.nodes, &Yog.Model.node(graph, &1))
Properties & Checks
Reachability
Check connectivity and extract components.
# Path existence check
Yog.Connectivity.reachable?(graph, 1, 10)

# Strongly Connected Components
Yog.Connectivity.strongly_connected_components(graph)
Invariants & Metrics
Verify structural properties and calculate health.
# DAG & Bipartite checks
Yog.Property.Cyclicity.is_dag?(graph)
Yog.Property.Bipartite.bipartite?(graph)

# Walk-based analysis
Yog.Traversal.fold_walk(over: graph, from: 1, with: callback)
	See Yog.Traversal for lexicographical and acyclic-guaranteed walks.

Algorithms & Metrics
Centrality & Rankings
Determine the relative importance of nodes within a network. 
	Betweenness: Nodes that act as bridges.
	Closeness: Nodes that can reach others quickly.
	PageRank: Probability-based ranking.

# Measure node importance
Yog.Centrality.Brandes.betweenness_centrality(graph)

# Direct neighbor counts
Yog.Model.successors(graph, 1)
Yog.Model.predecessors(graph, 1)
	See Yog.Centrality for advanced sorting and weighting.

Network Health
Measure the global structural quality and efficiency of the graph.
	Diameter: Longest shortest path (max separation).
	Radius: Min eccentricity (best central hub).
	Assortativity: Do hubs connect to hubs or leaves?

# Check graph compactness
Yog.Health.diameter(graph)
Yog.Health.radius(graph)

# Social (positive) vs Bio (negative) patterns
assort = Yog.Health.assortativity(graph)
	Use Yog.Health for average path lengths and eccentricity.

Community Detection
Partition nodes into densely connected modules (clusters).
	Louvain/Leiden: Fast, quality-guaranteed modularity optimization.
	Label Propagation: Linear-time, near-instant detection.
	Modularity (Q): Metric for "strength" of the grouping.

# Auto-detect clusters
res = Yog.Community.Louvain.detect(graph)

# Get metrics
Yog.Community.modularity(graph, res)
Yog.Community.sizes(res) # %{comm_id => count}
	See Yog.Community for hierarchical and local detection.

Minimum Spanning Trees (MST)
Find a subset of edges that connects all nodes with the minimum (or maximum) total weight.
	Prim/Kruskal: Standards for minimizing edge weight.
	Maximum MST: Connect nodes using the heaviest edges (for redundancy/capacity).
	Uniform Forest: Sample a random spanning tree from all possible trees.

# Standard Minimum Tree (Kruskal)
Yog.MST.kruskal(graph)

# Maximum Spanning Tree (MaxST)
Yog.MST.kruskal_max(graph)
Yog.MST.prim_max(graph, from: 1)

# Random Spanning Forest (Wilson's)
Yog.MST.uniform_spanning_tree(graph)
	Refer to Yog.MST for algorithm selection and weighted/unweighted variations.

Connectivity & SCC
Analyze how nodes are grouped together in directed and undirected graphs.
	SCC (Strongly Connected): For directed graphs (Tarjan/Kosaraju).
	Critical Points: Bridges and Articulation Points.

# Find all Strongly Connected Components
Yog.Connectivity.scc(directed_graph)

# Analyze bridges and articulation points
results = Yog.Connectivity.analyze(undirected_graph)
results.bridges             # [{1, 2}, {3, 4}]
results.articulation_points  # [2, 5]
	See Yog.Connectivity for k-core and weakly connected components.

DAG
Type Safety
Validate and wrap a directed graph as a DAG to guarantee it has no cycles.
# Wrap a graph and prove acyclicity
{:ok, dag} = Yog.DAG.from_graph(graph)

# Convert back to general graph
Yog.DAG.to_graph(dag)
	Use Yog.DAG.Model for operations that maintain DAG invariants.

Total Functions
Specialized algorithms that are guaranteed to succeed on a DAG without checking for cycles internally.
# Linear time topological sort (Infallible)
sorted_nodes = Yog.DAG.topological_sort(dag)

# Find the longest path (Critical Path analysis)
path = Yog.DAG.longest_path(dag)
	See Yog.DAG.Algorithm for more DAG-specific logic.

Maze & Grid
Grid Construction
Build 2D structures from nested lists with movement constraints.
	Topologies: rook (4-way), queen (8-way), bishop (diagonal).
	Predicates: walkable (only specific values), avoiding (all but values).

maze_data = [
  [".", ".", "#"],
  [".", "#", "."],
  [".", ".", "G"]
]

# Create an undirected grid where "." is walkable
grid = Yog.Builder.Grid.from_2d_list(
  maze_data, 
  :undirected, 
  Yog.Builder.Grid.walkable(".")
)

# Convert to standard graph for pathfinding
graph = Yog.Builder.GridGraph.to_graph(grid)
	Use Yog.Builder.Grid for coordinate-to-node-ID mapping.

Navigation & Heuristics
Estimate costs between grid cells for informed search (A*).
	Manhattan: Sum of cardinal steps (for Rook).
	Chebyshev: Max of row/col diffs (for Queen/8-way).
	Octile: Precise 8-way cost using √2 for diagonals.

# Convert coordinates back and forth
id = Yog.Builder.Grid.coord_to_id(row, col, cols)
{r, c} = Yog.Builder.Grid.id_to_coord(id, cols)

# Distance estimate for A*
heuristic = fn a, b -> 
  Yog.Builder.Grid.manhattan_distance(a, b, cols)
end
	Refer to Yog.Builder.Grid for all distance formulas.

Maze Generation
Create procedurally generated "perfect mazes" (spanning trees on a grid).
	Recursive Backtracker: Classic, twisty corridors.
	Prim: Radial texture, many dead ends.
	Kruskal: Well-balanced, uniform look.

# Generate a 20x20 twisty maze
maze = Yog.Generator.Maze.recursive_backtracker(20, 20, seed: 123)

# Render to string for CLI output
IO.puts(Yog.Render.ASCII.grid_to_string(maze))
	See Yog.Generator.Maze for algorithm selection guide.

IO & Serialization
Constructors
Import graphs from standard formats or data structures.
# From adjacency lists/matrices
Yog.from_adjacency_list(:directed, [{"A", ["B", "C"]}])

# Parse external formats
Yog.IO.JSON.decode(json)
Yog.IO.GraphML.decode(xml)
Export & Render
Serialize graphs for storage or visualization.
# Export to basic list
Yog.to_adjacency_list(graph)

# Render to diagram strings
Yog.Render.Dot.render(graph)
Yog.Render.Mermaid.render(graph)
	See Yog.IO for LEDA, Pajek, and GDF support.

Documentation & Navigation
Browser Shortcuts
	s: Open quick-search bar
	g: Go to Hex package docs
	?: Toggle shortcut help
	/: Focus search bar

Linking Syntax (Backticks)
	`Module`: Link to a module
	`mod.fun/2`: Link to a function
	`t:mod.type/0`: Link to a type
	`c:mod.fun/1`: Link to a callback
	`[txt](`fun/1`)`: Custom link text

Yog is a set of Graph and Network algorithms and data structures implemented in Elixir and packaged as a common API.
There is also a similar library written in Gleam, which I used to learn Gleam by collecting and translating most of the algorithms I have written in Elixir over the years. 
Both libraries are actively maintained, with the Elixir version getting new features more frequently, and also sporting more experimental features.
[!WARNING]
API Stability: Until the version reaches 1.0.0, there may be some hiccups in performance, and documentation quality may be less than optimal. Before version 0.99.0 is released in December 2026, the primary focus is on performance, documentation, and bugfixes. From v0.97.0 onwards, there will be no breaking changes until v2.0 (no plans for that).

Features
YogEx provides comprehensive graph algorithms organized into modules:
Core Capabilities
Pathfinding & Flow — Shortest paths (Dijkstra, A*, Bellman-Ford, Floyd-Warshall, Johnson's), maximum flow (Edmonds-Karp), min-cut (Stoer-Wagner), and implicit state-space search for on-demand graphs.
Network Analysis — Centrality measures (PageRank, betweenness, closeness, eigenvector, Katz), community detection (Louvain, Leiden, Infomap, Walktrap), and network health metrics.
Connectivity & Structure — SCCs (Tarjan/Kosaraju), bridges, articulation points, K-core decomposition, and reachability analysis with exact and HyperLogLog-based estimation.
Graph Operations — Union, intersection, difference, Cartesian product, power, isomorphism, and O(1) transpose.
Developer Experience
Generators & Builders — Classic patterns (complete, cycle, grid, Petersen), random models (SBM, R-MAT), and a comprehensive Maze Generation suite (Recursive Backtracker, Wilson's, Kruskal's, Eller's, etc.) with labeled and grid builders.
I/O & Visualization — GraphML, GDF, Pajek, LEDA, TGF, JSON serialization plus ASCII, DOT, and Mermaid rendering.
Functional (Experimental) — Pure inductive graph library (FGL) for elegant recursive algorithms.
Complete Algorithm Catalog — See all 60+ algorithms, underlying data structures (Pairing Heap, Union-Find, HyperLogLog), and selection guidance with Big-O complexities.
Installation
Basic Installation
Add YogEx to your list of dependencies in mix.exs:
def deps do
  [
    {:yog_ex, "~> 0.98.0"}
  ]
end
Then run:
mix deps.get

Optional Dependencies
YogEx includes several optional dependencies that enable additional I/O and interoperability features:
	Dependency	Module	Purpose
	:saxy	Yog.IO.GraphML	Fast streaming XML parser for GraphML files (3-4x faster than default :xmerl)
	:jason	Yog.IO.JSON	JSON serialization/deserialization for D3.js, Cytoscape, vis.js, NetworkX formats
	:libgraph	Yog.IO.Libgraph	Bidirectional conversion with libgraph library

To use these features, add the optional dependencies to your mix.exs:
def deps do
  [
    {:yog_ex, "~> 0.98.0"},
    {:saxy, "~> 1.5"},       # For fast GraphML/XML parsing
    {:jason, "~> 1.4"},      # For JSON import/export
    {:libgraph, "~> 0.16"}   # For libgraph interoperability
  ]
end
XML/GraphML with Saxy
# Reading large GraphML files is significantly faster with saxy
{:ok, graph} = Yog.IO.GraphML.read("large_network.graphml")

# Writing GraphML
Yog.IO.GraphML.write("output.graphml", graph)
JSON Serialization with Jason
# Export to various JSON formats
json = Yog.IO.JSON.to_json(graph, Yog.IO.JSON.export_options_for(:d3_force))

# Import from JSON
{:ok, graph} = Yog.IO.JSON.from_json(json_string)
Libgraph Interoperability
# Convert Yog graph to libgraph
libgraph = Yog.IO.Libgraph.to_libgraph(graph)

# Convert libgraph back to Yog
{:ok, yog_graph} = Yog.IO.Libgraph.from_libgraph(libgraph)
Livebook
For livebook, add the following:
Mix.install(
  {:yog_ex, "~> 0.98.0"}
)
There is a Kino App that can be used to explore the library and create and render graphs.
Usage
alias Yog.Pathfinding

# Create a directed graph
graph =
  Yog.directed()
  |> Yog.add_node(1, "Start")
  |> Yog.add_node(2, "Middle")
  |> Yog.add_node(3, "End")
  |> Yog.add_edge_ensure(from: 1, to: 2, with: 5)
  |> Yog.add_edge_ensure(from: 2, to: 3, with: 3)
  |> Yog.add_edge_ensure(from: 1, to: 3, with: 10)

# Find shortest path using Dijkstra (uses :ok/:error tuples and Path struct)
case Pathfinding.shortest_path(
  in: graph,
  from: 1,
  to: 3
) do
  {:ok, path} ->
    IO.puts("Found path with weight: #{path.weight}")
  :error ->
    IO.puts("No path found")
end
# => Found path with weight: 8
Examples
Detailed examples are located in the examples/README.md file.
Advent of Code Solutions
YogEx is used to solve Advent of Code challenges. 
See all Advent of Code solutions tagged with graph that demonstrate usage of YogEx algorithms in the Advent of Code repository.
Projects Using Yog
Yog powers the following open-source libraries that build domain-specific abstractions on top of its graph engine:
Choreo — Domain-Specific Diagram Builders
Analysis-first diagramming for Elixir. Instead of drawing static pictures, you model systems and get live answers — reachability, cycles, bottlenecks, threat generation, and more.
	Choreo — Infrastructure architecture diagrams (databases, caches, services, queues)
	Choreo.FSM — Finite state machines with determinism checks and shortest accepting paths
	Choreo.Dataflow — Pipeline diagrams with throughput simulation and backpressure detection
	Choreo.Dependency — Software dependency graphs with cycle detection and layer enforcement
	Choreo.DecisionTree — Classification trees with feature importance and pruning
	Choreo.MindMap — Concept mapping with orphan detection and root-to-leaf paths
	Choreo.ThreatModel — STRIDE threat modeling with auto-generated severity scoring
	Choreo.Workflow — Task orchestration with critical-path analysis and Saga-pattern compensations

Tapestry — Graph-Native Domain Engine
Model structured domains as typed multigraphs. Kanban boards, timelines, dependency networks, and structural analysis are all projections of the same underlying graph.
	Project management with milestones, tasks, users, and labels
	Query what blocks a task, what's ready to start, and who's the bottleneck
	Critical-path analysis and transitive dependency tracking
	Renders natively to Mermaid for GitHub, Notion, and Obsidian

Meridian — Spatial Graphs
Projection-aware spatial graphs for Elixir. Brings geography into graph theory with coordinate-reference-system safety, map ingestion, and spatial algorithms.
	Build graphs from H3 hex grids and geohash rectangles
	Ingest and render GeoJSON for road networks and geographic data
	Spatially-informed A*, Dijkstra, and widest-path routing
	CRS-aware edge weights with earth-distance heuristics

Development
Running Tests
mix test

Run tests for a specific module:
mix test test/yog/pathfinding/dijkstra_test.exs

Project Structure
	lib/yog/ — Core graph library modules (pure Elixir)
	test/ — Unit tests and doctests
	examples/ — Real-world usage examples

Property-Based Testing
This library uses property-based testing (PBT) via StreamData to ensure that algorithms hold up against a wide range of automatically generated graph structures. 
See the PROPERTIES.md for a complete catalog of all algorithmic invariants (hypotheses) verified by the test suite.
AI Assistance
Parts of this project were developed with the assistance of AI coding tools. All AI-generated code has been reviewed, tested, and validated by the maintainer.

This directory contains practical examples demonstrating various features of the YogEx graph library.
Table of Contents
	Pathfinding - Shortest path algorithms
	Community Detection - Finding clusters in graphs
	Graph Generators - Creating graphs programmatically
	Grid Builder - Maze solving with 2D grids
	Labeled Graph Builder - Human-readable node labels
	Centrality Analysis - Measuring node importance
	Graph I/O - Importing and exporting graphs
	Functional Graphs - Pure functional graph operations
	Running Examples


Pathfinding
Find the shortest path between nodes using Dijkstra's algorithm.
alias Yog.Pathfinding

# Create a directed graph
graph =
  Yog.directed()
  |> Yog.add_node(1, "Start")
  |> Yog.add_node(2, "Middle")
  |> Yog.add_node(3, "End")
  |> Yog.add_edge_ensure(from: 1, to: 2, with: 5)
  |> Yog.add_edge_ensure(from: 2, to: 3, with: 3)
  |> Yog.add_edge_ensure(from: 1, to: 3, with: 10)

# Find shortest path using Dijkstra (uses :ok/:error tuples and Path struct)
case Pathfinding.shortest_path(
  in: graph,
  from: 1,
  to: 3
) do
  {:ok, path} ->
    IO.puts("Found path with weight: #{path.weight}")
  :error ->
    IO.puts("No path found")
end
# => Found path with weight: 8
Related file: gps_navigation.exs

Community Detection
Identify natural clusters in a social network using the Louvain algorithm.
alias Yog.Community

# Build a graph with two communities
graph =
  Yog.undirected()
  |> Yog.add_edge_with(1, 2, 1, & &1)
  |> Yog.add_edge_with(2, 3, 1, & &1)
  |> Yog.add_edge_with(1, 3, 1, & &1)   # Triangle: 1-2-3
  |> Yog.add_edge_with(4, 5, 1, & &1)
  |> Yog.add_edge_with(5, 6, 1, & &1)
  |> Yog.add_edge_with(4, 6, 1, & &1)   # Triangle: 4-5-6
  |> Yog.add_edge_with(3, 4, 1, & &1)   # Bridge between communities

communities = Community.Louvain.detect(graph)
IO.puts("Found #{communities.num_communities} communities")
# => Found 2 communities

modularity = Community.modularity(graph, communities)
IO.puts("Modularity: #{modularity}")
# => Modularity: 0.35714285714285715
Related files: zacharys_karate_club.exs, social_network_analysis.exs

Graph Generators
Create common graph patterns and random graphs for testing and benchmarking.
alias Yog.Generator.{Classic, Random}

# Classic graph patterns
complete = Classic.complete(10)       # K₁₀ complete graph
cycle = Classic.cycle(20)              # C₂₀ cycle graph
petersen = Classic.petersen()           # The famous Petersen graph
grid = Classic.grid_2d(5, 5)           # 5×5 grid lattice

# Random graph models
sparse = Random.erdos_renyi_gnp(100, 0.05)    # G(n,p) model
scale_free = Random.barabasi_albert(1000, 3)   # Preferential attachment
small_world = Random.watts_strogatz(100, 6, 0.1)  # Small-world
Related file: graph_generation_showcase.exs

Grid Builder
Solve mazes and pathfinding problems on 2D grids.
alias Yog.Builderr.Grid
alias Yog.Pathfinding
alias Yog.Render.ASCII

# Build a maze from a 2D grid
maze = [
  [".", "#", "#", "."],
  [".", ".", "#", "#"],
  ["#", ".", ".", "."],
  ["#", "#", "#", "."],
  ["#", ".", "#", "."],
]

# Create grid with walkable predicate
grid = Grid.from_2d_list(maze, :undirected, Grid.including(["."]))

IO.puts(ASCII.grid_to_string(grid, %{0 => "S", 19 => "E"}))

# Prints the Maze:
#
#  +---+---+---+---+
#  | S |   |   |   |
#  +   +---+---+---+
#  |       |   |   |
#  +---+   +---+---+
#  |   |           |
#  +---+---+---+   +
#  |   |   |   |   |
#  +---+---+---+   +
#  |   |   |   | E |
#  +---+---+---+---+
#
Related file: maze_solver.exs

Labeled Graph Builder
Work with human-readable labels instead of integer IDs.
alias Yog.Builder.Labeled
alias Yog.Pathfinding

# Build graphs with meaningful labels instead of integer IDs
builder =
  Labeled.directed()
  |> Labeled.add_edge("London", "Paris", 450)
  |> Labeled.add_edge("Paris", "Berlin", 878)
  |> Labeled.add_edge("London", "Berlin", 930)

# Convert to graph for algorithms
graph = Labeled.to_graph(builder)

# Look up internal IDs by label
{:ok, london_id} = Labeled.get_id(builder, "London")
{:ok, berlin_id} = Labeled.get_id(builder, "Berlin")

# Find shortest path using integer IDs
case Pathfinding.shortest_path(in: graph, from: london_id, to: berlin_id) do
  {:ok, path} ->
    labeled_path = Enum.map(path.nodes, fn id ->
      graph.nodes[id]
    end)
    IO.puts("Route: #{Enum.join(labeled_path, " -> ")}, Distance: #{path.weight} km")
end
# => Route: London -> Berlin, Distance: 930 km
Related file: gps_navigation.exs

Centrality Analysis
Measure the importance of nodes in a network.
alias Yog.Centrality

graph =
  Yog.directed()
  |> Yog.add_node(1, nil) 
  |> Yog.add_node(2, nil) 
  |> Yog.add_node(3, nil)
  |> Yog.add_edges!([{1, 2, 1}, {2, 3, 1}, {1, 3, 1}])

# Various centrality measures
pagerank = Centrality.pagerank(graph)
# => %{1 => 0.19757597883790196, 2 => 0.2815434776603495, 3 => 0.5208805435017486}
betweenness = Centrality.betweenness(graph)
# => %{1 => 0.0, 2 => 0.0, 3 => 0.0}
closeness = Centrality.closeness(graph)
# => %{1 => 1.0, 2 => 0.0, 3 => 0.0}
degree = Centrality.degree(graph, :out_degree)
# => %{1 => 1.0, 2 => 0.5, 3 => 0.0}
Related file: zacharys_karate_club.exs

Graph I/O
Import and export graphs in various formats.
alias Yog.IO.{GDF, GraphML, Pajek}

# Create graph
graph =
  Yog.directed()
  |> Yog.add_node(1, "Alice")
  |> Yog.add_node(2, "Bob")
  |> Yog.add_edge_ensure(from: 1, to: 2, with: "follows")

# Serialize to popular formats like GraphML, TGF, LEDA, Pajek, JSON, or GDF
gdf_string = GDF.serialize(graph)
graphml_string = GraphML.serialize(graph)
pajek_string = Pajek.serialize(graph)

# Parse from string or read from file
{:ok, graph} = GraphML.deserialize(graphml_string)

# Read directly from file
# {:ok, loaded} = GraphML.read("slashdot.xml")
Related files: render_dot.exs, render_json.exs, render_mermaid.exs

Functional Graphs
Use pure functional graph operations with inductive decomposition (Experimental).
alias Yog.Functional.{Algorithms, Model}

# Create an inductive functional graph
graph =
  Model.empty()
  |> Model.put_node(1, "A")
  |> Model.put_node(2, "B")
  |> Model.put_node(3, "C")
  |> Model.add_edge!(1, 2, 1)
  |> Model.add_edge!(2, 3, 2)

# Inductive Top-Sort - consumes the graph naturally, no mutable visited sets!
{:ok, order} = Algorithms.topsort(graph)
# => [1, 2, 3]

# Inductive Dijkstra
Algorithms.shortest_path(graph, 1, 3)
# => {:ok, [1, 2, 3], 3}
See the Functional Graphs README for a deep dive into the build/burn pattern, context-based traversal, and the philosophy of inductive graph decomposition.
Related file: functional_example.exs

Running Examples
To run any example:
# From the project root
mix run examples/gps_navigation.exs
mix run examples/network_bandwidth.exs
mix run examples/maze_solver.exs
# etc.

Available Example Files
	File	Description
	gps_navigation.exs	Shortest path with labeled cities using Dijkstra
	network_bandwidth.exs	Maximum flow for network optimization
	social_network_analysis.exs	Finding strong components in social graphs
	zacharys_karate_club.exs	Louvain community detection on Karate Club dataset
	maze_solver.exs	Solving a grid maze using BFS/DFS
	maze_solver_directional.exs	Solving a grid maze with directional constraints
	graph_creation.exs	Manually building and querying graphs
	graph_generation_showcase.exs	Showcasing classic generators (cliques, cycles, grids)
	functional_example.exs	Pure functional graph operations (experimental)
	render_dot.exs	Exporting graphs to Graphviz DOT format
	render_json.exs	Serializing/deserializing graphs to JSON
	render_mermaid.exs	Rendering graphs using Mermaid.js syntax
	ascii_maze.exs	Generating and printing ASCII mazes
	bridges_of_konigsberg.exs	Modeling the Seven Bridges of Königsberg
	cave_path_counting.exs	Custom backtracking DFS path counting
	city_distance_matrix.exs	Shortest paths matrix on city networks
	flood_fill.exs	BFS-based grid flood fill example
	global_min_cut.exs	Computing global min-cut on flow networks
	job_assignment.exs	Bipartite matching for job assignment
	job_matching.exs	Stable marriage/matching with Gale-Shapley
	medical_residency.exs	Gale-Shapley matching for residency matching
	maze_gallery.exs	Maze generator comparison and generation
	network_cable_layout.exs	Spanning trees for cable layouts
	number_of_islands.exs	BFS-based grid island counting
	task_ordering.exs	Resolving dependencies with topological sort
	task_scheduling.exs	Critical path method for scheduling


For more information, see the main README or the API documentation.


  

    Yog.Functional

Functional inductive graphs for Elixir researchers and learners.

This package provides an inductive graph representation based on Martin Erwig's Functional Graph Library (FGL). Unlike the traditional adjacency-list approach used in Yog.Graph, these graphs are defined recursively, enabling algorithms to be written as pure, recursive functions without explicit "visited" sets.
Philosophy: The Inductive Principle
An inductive graph is defined by two constructors:
	 Empty: A graph with no nodes.
	 Embed: Formed by adding a node context to an existing graph. This is the inverse of match/2.

The Inductive Duality: Match and Embed
	  Match (Decompose): Extracts a node and its context. The resulting remaining graph is the new "shrunken" world view. The node and all its incident edges are gone, so the graph is structurally smaller.
	  Embed (Compose): The inverse. It takes a context and a graph to produce a new (larger) graph.

alias Yog.Functional.Model

# Match node 1
{:ok, context, shrunken_graph} = Model.match(graph, 1)

# Embed it back (potentially into a transformed graph)
new_graph = Model.embed(shrunken_graph, context)
Pattern Matching Friendly
Just as lists are defined as [head | tail], these graphs are defined as context & remaining. Every recursive step handles these as "head" and "tail", allowing for code that reads like pattern matching:
	 Base case: Model.empty() (the empty graph).
	 Recursive case: match(graph, id) gives you the context (head) and the shrunken world view (tail).

Every time you recurse with the shrunken_graph, you are operating on a self-contained "sub-universe" where the earlier nodes simply do not exist.
By working with remaining_graph, algorithms naturally prevent revisits: once a node is matched, it and all its edges are gone from the perspective of the recursion. The algorithm terminates when Model.empty?(graph) is true.
Data Example: Shrinking the Graph
Consider a simple directed graph: 1 -> 2 -> 3.
	 Initial Graph: Represents the full structure {1, 2, 3}.
	 Match 1: 	  Context: id: 1, label: "A", out_edges: %{2 => nil}
	  Remaining: A graph containing only 2 -> 3. Node 1 and the edge 1 -> 2 have been structurally removed.


	 Match 2 (on the remaining graph):	  Context: id: 2, label: "B", out_edges: %{3 => nil}
	  Remaining: A graph containing only the isolated node 3.


	 Match 3:	  Context: id: 3, label: "C"
	  Remaining: Model.empty().



The Build/Burn Pattern: Context as State
The fundamental pattern for working with inductive graphs is what we call build/burn — alternating between deconstructing (burning) and reconstructing (building) the graph. The context is your handle on the current "focus" of the graph.
Deconstruction (Burn): match/2
When you match a node, you receive:
	The Context — the node's ID, label, and all its incident edges
	The Remaining Graph — the graph with this node burned away, edges and all

alias Yog.Functional.Model

graph = Model.empty()
|> Model.put_node(1, "Alice")
|> Model.put_node(2, "Bob")
|> Model.put_node(3, "Carol")
|> Model.add_edge!(1, 2, :follows)
|> Model.add_edge!(2, 3, :follows)

# Burn node 1 — extract it from the graph
{:ok, ctx, remaining} = Model.match(graph, 1)

# ctx.id == 1
# ctx.label == "Alice"
# ctx.out_edges == %{2 => :follows}
# ctx.in_edges == %{} (no one follows Alice in this graph)

# remaining contains nodes 2 and 3, but NO edge from 1->2
# because node 1 and all its edges have been burned away
Construction (Build): embed/2
The inverse operation restores a context into a graph. This enables transform-then-rebuild workflows:
# Transform the context (e.g., increment a counter in the label)
new_ctx = %{ctx | label: %{ctx.label | visits: ctx.label.visits + 1}}

# Build it back into a (possibly different) graph
restored_graph = Model.embed(new_ctx, remaining)
Traversal Example: Finding All Nodes with a Property
Here's how you traverse by burning through the graph, collecting matches:
def find_influencers(graph, min_followers, acc \\ []) do
  # Try to match any remaining node
  case Model.match_any(graph) do
    {:error, :empty} ->
      # Base case: graph is burned away completely
      Enum.reverse(acc)
      
    {:ok, ctx, remaining} ->
      # ctx is our "current state" — we have full info about this node
      follower_count = map_size(ctx.in_edges)
      
      # Decide based on context, then continue burning the rest
      new_acc = if follower_count >= min_followers do
        [{ctx.id, ctx.label, follower_count} | acc]
      else
        acc
      end
      
      # Recurse on the BURNED graph — ctx.id is gone, so we can't revisit
      find_influencers(remaining, min_followers, new_acc)
  end
end

# Usage:
# find_influencers(social_graph, 1000, [])
# => [{2, "Bob", 1500}, {5, "Eve", 2300}]
Key Insight: The Context is Your Iterator
In traditional graphs, you iterate over node IDs and look up data. In FGL:
	The context bundles identity, data, AND connectivity — you have everything needed to decide "what next" without additional lookups
	The remaining graph IS your visited set — burned nodes simply don't exist anymore
	No back-edges possible — once you match a node, it cannot be reached from any future recursive call

This makes algorithms like DFS trivially correct without explicit cycle detection:
def dfs(graph, []), do: []
def dfs(graph, [v | vs]) do
  case Model.match(graph, v) do
    {:ok, ctx, remaining} ->
      # ctx.out_edges tells us where to go next
      # remaining ensures we never come back to ctx.id
      neighbors = Map.keys(ctx.out_edges)
      [ctx | dfs(remaining, neighbors ++ vs)]
      
    {:error, :not_found} ->
      # Already burned — skip
      dfs(graph, vs)
  end
end
Features
	  Memory Efficiency: Leverages Elixir's persistent data structures (Maps) to represent the recursive decomposition efficiently.
	  Algorithm Suite: Includes Topological Sort, SCC (Kosaraju), Dijkstra, Prim's MST, and Dominators.
	  Interop: Convert to and from the adjacency-based Yog.Graph model using Model.from_adjacency_graph/1.

Modules
	  Yog.Functional.Model — Core inductive graph type, match/2, embed/2, empty/0, and conversion to/from Yog.Graph.
	  Yog.Functional.Algorithms — Classic algorithms: topological sort, strongly connected components (Kosaraju), Dijkstra, Prim's MST, and dominators.
	  Yog.Functional.Analysis — Graph analysis utilities built on the inductive representation.
	  Yog.Functional.Traversal — Pure functional traversals (DFS, BFS) using the match-and-burn pattern.
	  Yog.Functional.Transform — Graph transformations: map, filter, fold, and structural rewrites over inductive graphs.

References
	  Martin Erwig (2001): Inductive Graphs and Functional Graph Algorithms - The foundational paper.
	  Haskell FGL: hackage.haskell.org/package/fgl - The industry-standard functional graph library.
	  Programming in Haskell (Graham Hutton): Chapter on functional graph algorithms.


Note: This module is primarily for research and educational purposes. For high-performance production workloads involving millions of edges, the ephemeral Yog.Graph is generally recommended.


  

    Getting Started with Yog

Mix.install([
  {:yog_ex, "~> 0.98"},
  {:kino_vizjs, "~> 0.8.0"}
])
What is Yog?
যোগ (jōg) means connection, link, or union.
Yog is a comprehensive graph algorithm library for Elixir. It provides efficient, immutable data structures and a wide array of algorithms for network analysis, pathfinding, and community detection.
Creating your first Graph
In Yog, graphs are immutable structures. You can create an empty graph of a specific kind:
# Create a directed graph
g = Yog.directed()

# Create an undirected graph
u = Yog.undirected()
By default, a graph consists of:
	kind: :directed or :undirected
	nodes: A map of node_id => data
	edges: Adjacency maps for fast lookups and O(1) transpose.

Growing the Graph
We build graphs by adding nodes and edges. Since Yog is functional and immutable, every operation returns a new graph.
Adding Nodes
Nodes can have any term as an ID and any term as data.
g = 
  Yog.directed()
  |> Yog.add_node(1, %{label: "Start"})
  |> Yog.add_node(2, %{label: "End"})
  |> Yog.add_nodes_from([3, 4, 5]) # Adding multiple nodes with nil data
Now let's visualize the nodes. Yog supports both Dot and Mermaid for rendering graphs.
Kino.Layout.tabs(
  Dot: Kino.VizJS.render(Yog.Render.DOT.to_dot(g), height: "100px"),
  Mermaid: Kino.Mermaid.new(Yog.Render.Mermaid.to_mermaid(g))
)
Adding Edges
Edges connect nodes. In Yog, we provide several ways to add edges, depending on whether you want to ensure the nodes exist or handle missing nodes.
# 1. add_edge! - Raises if nodes don't exist
g = g |> Yog.add_edge!(1, 2, 10)
IO.inspect(g)

# 2. add_edge_ensure - Automatically creates nodes if they are missing
g = g |> Yog.add_edge_ensure(2, 3, 5, %{label: "Auto-created"})
IO.inspect(g)

# 3. add_simple_edge - Adds an edge with weight 1
g = g |> Yog.add_simple_edge!(3, 1)
IO.inspect(g)
And this is how the graph looks like:
Kino.Layout.tabs(
  Dot: Kino.VizJS.render(Yog.Render.DOT.to_dot(g)),
  Mermaid: Kino.Mermaid.new(Yog.Render.Mermaid.to_mermaid(g))
)
LabeledBuilder and the Builder pattern
Yog lets you define your graph with any data as node ID or edge weight.
graph =
  Yog.directed()
  |> Yog.add_node({0, 0}, nil)
  |> Yog.add_node(:point, %{})
  |> Yog.add_edge!({0, 0}, :point, :unreachable)


Yog.all_edges(graph)
However, many algorithms or graph file formats are simplier when the Node IDs are integers. It also makes traversals and lookups faster. The graph above could look like:
graph =
  Yog.directed()
  |> Yog.add_node(1, {{0, 0}, nil})
  |> Yog.add_node(2, {:point, %{}})
  |> Yog.add_edge!(1, 2, :unreachable)

Yog.all_edges(graph)
The example above is much cleaner. But it also come with an ergonomic tax. If consuming from a stream, one would need to keep a map of ID to Data, and Data to ID.
This is exactly where the LabeledBuilder becomes relevant.
Yog has a set of Builders that acts like a domain specific graph creation API - hides the implementation detail but creates Yog graph underneath. They can be added anywhere, but Yog provides a few builders to get you going.
The one we discuss here is the LabeledBuilder. This lets you build graphs with any data as NodeID but creates a graph with integer based ID.
Here's an example:
builder =
  Yog.Builder.Labeled.directed()
  |> Yog.Builder.Labeled.add_edge("A", "B", 10)
  |> Yog.Builder.Labeled.add_edge("B", "C", 5)

IO.puts("These are the labels")
IO.inspect(builder |> Yog.Builder.Labeled.all_labels())

IO.puts("A registry that maps labels to Node IDs")
IO.inspect(builder |> Yog.Builder.Labeled.to_registry())

IO.puts("The underlying graph")
graph = builder |> Yog.Builder.Labeled.to_graph()
IO.inspect(Yog.node_ids(graph))

IO.puts("To query between NodeID and Label")
IO.inspect(Yog.Builder.Labeled.get_id(builder, "A"))
IO.inspect(Yog.Builder.Labeled.get_label(builder, 0))

There are a few other builders:
	LiveBuilder - superpowered LabeledBuilder instead of creating the whole builder, it lets you continuously add/remove nodes from builder and create graph via a WAL log lite.
	GridGraph &amp; ToroidalGraph - Creates graphs from 2D grid data. Toroidal wraps ends of rows and columns.

A use case for Yog builders is Choreo where builder pattern is used to add system design DSL on top of Yog.
Examining the Graph
You can query the graph's structure using functions in the main Yog module or Yog.Model.
IO.puts "Nodes: #{Yog.node_count(g)}"
IO.puts "Edges: #{Yog.edge_count(g)}"

# Get successors of node 2
IO.inspect(Yog.successors(g, 2), label: "Successors of 2")

# Get neighbors regardless of direction
IO.inspect(Yog.neighbors(g, 3), label: "Neighbors of 3")

# Check if the graph is cyclic
IO.puts("Is cyclic? #{Yog.cyclic?(g)}")
Transformations
Yog excels at functional transformations. You can map or filter nodes and edges to create new graph versions.
# Double all edge weights
high_weight_graph = Yog.Transform.map_edges(g, fn weight -> 
  if is_number(weight), do: weight * 2, else: weight 
end)

# Filter for nodes with numeric IDs
numeric_only = Yog.Transform.filter_nodes_indexed(g, fn id, _ -> is_integer(id) end)
Comprehensive Algorithms
Yog comes with 223 registered algorithms, generators, and helper tools categorized across 22 areas
Pathfinding
cols = 6
grid_data = List.duplicate(List.duplicate(".", cols), cols)
grid_builder = Yog.Builder.Grid.from_2d_list(grid_data, :undirected, Yog.Builder.Grid.always())
graph = Yog.Builder.Grid.to_graph(grid_builder)

start_node = Yog.Builder.Grid.coord_to_id(0, 0, cols)
end_node = Yog.Builder.Grid.coord_to_id(5, 5, cols)

{:ok, path} = Yog.Pathfinding.Dijkstra.shortest_path(graph, start_node, end_node)

# Create base options with coordinate labels for nodes, and hide edge labels
base_opts = 
  Yog.Render.DOT.default_options()
  |> Map.put(:node_label, fn id, _data ->
    {r, c} = Yog.Builder.Grid.id_to_coord(id, cols)
    "#{r},#{c}"
  end)
  |> Map.put(:edge_label, fn _weight -> "" end)

# Pass base_opts to path_to_options to highlight path and retain custom labeling
opts = Yog.Render.DOT.path_to_options(path, base_opts)

Kino.VizJS.render(Yog.Render.DOT.to_dot(graph, opts), height: "800px")

Community Detection
# Generate a graph with 24 nodes, 3 communities, p_in = 0.7, p_out = 0.1
sbm = Yog.Generator.Random.sbm(24, 3, 0.7, 0.1, community_sizes: [8, 8, 8])
result = Yog.Community.Louvain.detect(sbm)

IO.puts("Detected #{result.num_communities} communities")
IO.inspect(result.assignments, label: "Node -> Community")

# Visualize with community colors
opts = Yog.Render.DOT.community_to_options(result)
Kino.VizJS.render(Yog.Render.DOT.to_dot(sbm, opts), engine: "circo", height: "1200px")
Visualization Formats
Yog supports both Graphviz DOT and Mermaid.js output.
# Graphviz DOT (rich, customizable)
dot = Yog.Render.DOT.to_dot(g)

# Mermaid.js (great for Markdown docs)
mermaid = Yog.Render.Mermaid.to_mermaid(g)


IO.puts("DOT")
IO.puts(dot)
IO.puts("Mermaid")
IO.puts(mermaid)

Kino.Layout.tabs(
  Dot: Kino.VizJS.render(dot),
  Mermaid: Kino.Mermaid.new(mermaid)
)
Serialization
You can easily import/export graphs in various formats like GraphML, JSON, GDF, Graph6, or DOT.
xml = """
<?xml version="1.0" encoding="UTF-8"?>
<graphml xmlns="http://graphml.graphdrawing.org/xmlns"  
    xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"
    xsi:schemaLocation="http://graphml.graphdrawing.org/xmlns
     http://graphml.graphdrawing.org/xmlns/1.0/graphml.xsd">
  <graph id="G" edgedefault="undirected">
    <node id="n0"/>
    <node id="n1"/>
    <node id="n2"/>
    <node id="n3"/>
    <node id="n4"/>
    <node id="n5"/>
    <node id="n6"/>
    <node id="n7"/>
    <node id="n8"/>
    <node id="n9"/>
    <node id="n10"/>
    <edge source="n0" target="n2"/>
    <edge source="n1" target="n2"/>
    <edge source="n2" target="n3"/>
    <edge source="n3" target="n5"/>
    <edge source="n3" target="n4"/>
    <edge source="n4" target="n6"/>
    <edge source="n6" target="n5"/>
    <edge source="n5" target="n7"/>
    <edge source="n6" target="n8"/>
    <edge source="n8" target="n7"/>
    <edge source="n8" target="n9"/>
    <edge source="n8" target="n10"/>
  </graph>
</graphml>
"""

# Let's load the graph
{:ok, graph} = Yog.IO.GraphML.deserialize(xml)

# Export to GDF (GUESS format)
gdf = Yog.IO.GDF.serialize(graph)
IO.puts String.slice(gdf, 0, 300) <> "..."

Kino.Layout.grid([
  Kino.Markdown.new("""
  ```text
  #{gdf}
  ```
  """),
  graph
  |> Yog.Render.DOT.to_dot()
  |> Kino.VizJS.render(height: "600px")
], columns: 2)
Next Steps
Explore the Algorithm Catalog to see everything Yog can do!


  

    How-To: Maze Generation & Solving

Mix.install([
  {:yog_ex, "~> 0.98"},
  {:kino_vizjs, "~> 0.8.0"}
])
Introduction
A perfect maze is mathematically defined as a spanning tree over a grid graph. In a perfect maze:
	Every cell is reachable from every other cell.
	There is exactly one unique path between any two cells (no loops/cycles, and no isolated cells).

Yog provides generators for producing perfect mazes using various algorithms, along with builder utilities to convert these mazes into standard graphs that can be searched and solved using standard pathfinding algorithms.
This guide walks step-by-step through generating a maze, representing it as a graph, and solving it programmatically.

Step 1: Generate the Maze
First, we generate a maze. We'll use the Recursive Backtracker algorithm, which generates winding, complex passages with relatively long corridors.
rows = 12
cols = 12

# Generate a maze structure
maze = Yog.Generator.Maze.recursive_backtracker(rows, cols, seed: 42)

# Print a basic ASCII representation of the wall layouts
IO.puts(Yog.Render.ASCII.grid_to_string_unicode(maze))

Step 2: Convert the Grid to a Graph
Although the maze is structured as a grid, we need to convert it into a standard Yog.Graph to use Yog's graph traversal and pathfinding algorithms.
The Yog.Builder.GridGraph module handles this. It creates a node for each cell and adds undirected edges between adjacent cells if there is no wall between them.
# Convert the maze structure into a standard graph
graph = Yog.Builder.GridGraph.to_graph(maze)

# Inspect the graph properties
IO.inspect(Yog.node_count(graph), label: "Graph Node Count")
IO.inspect(Yog.edge_count(graph), label: "Graph Edge Count (Passages)")

Step 3: Define the Start, Exit, and Solve
In a grid graph, each node is indexed by a flat integer ID. We can translate 2D grid coordinates (row, col) into their corresponding node IDs using Yog.Builder.GridGraph.coord_to_id/3.
For our example:
	Start: Top-Left corner (0, 0)
	Exit: Bottom-Right corner (rows - 1, cols - 1)

Once we have the node IDs, we can use Yog.Pathfinding.shortest_path/1 to find the unique path from start to exit.
start_node = Yog.Builder.GridGraph.coord_to_id(maze, 0, 0)
exit_node = Yog.Builder.GridGraph.coord_to_id(maze, rows - 1, cols - 1)

# Find the path using Dijkstra's algorithm
case Yog.Pathfinding.shortest_path(in: graph, from: start_node, to: exit_node) do
  {:ok, path} ->
    IO.inspect(path.nodes, label: "Path Node IDs")
    IO.inspect(path.weight, label: "Path Length (Steps)")

  :error ->
    IO.puts("No path exists!")
end

Step 4: Annotate and Render the Solution
We can display the solved path by passing an "occupants" map to the ASCII renderer. The renderer will place our custom symbols inside the maze cells corresponding to the solved path.
{:ok, path} = Yog.Pathfinding.shortest_path(in: graph, from: start_node, to: exit_node)

# Create a map of cell IDs to unicode indicators
occupants =
  path.nodes
  # Mark path nodes with a bullet
  |> Map.new(fn id -> {id, "•"} end)
  # Mark starting point
  |> Map.put(start_node, "S")
  # Mark exit point
  |> Map.put(exit_node, "E")

# Render the final layout
IO.puts(Yog.Render.ASCII.grid_to_string_unicode(maze, occupants))

Step 5: Visualizing the Underlying Spanning Tree
To see how the graph looks visually, we can render it to DOT format and display it using Kino.VizJS. This clearly exposes the spanning tree structure.
# We'll use a small 6x6 maze to keep the diagram readable
small_maze = Yog.Generator.Maze.recursive_backtracker(6, 6, seed: 100)
small_graph = Yog.Builder.GridGraph.to_graph(small_maze)

# Render graph network using Graphviz
dot = Yog.Render.DOT.to_dot(small_graph)
Kino.VizJS.render(dot, height: "800px")

Step 6: Comparing Corridor "Textures"
Different maze generation algorithms create different structural patterns or "textures". For example, a Binary Tree algorithm has a strong diagonal bias with long corridors running east/south, while Wilson's algorithm generates completely uniform and organic corridor patterns.
Let's generate the same size grid using different algorithms, solve them, and compare their solved path lengths to see how layout textures impact navigation:
size = 20

algorithms = [
  {"Binary Tree", &Yog.Generator.Maze.binary_tree/3},
  {"Sidewinder", &Yog.Generator.Maze.sidewinder/3},
  {"Recursive Backtracker", &Yog.Generator.Maze.recursive_backtracker/3},
  {"Wilson's", &Yog.Generator.Maze.wilson/3}
]

for {name, generator} <- algorithms do
  # Generate with same dimensions and seed
  maze = generator.(size, size, seed: 42)
  g = Yog.Builder.GridGraph.to_graph(maze)
  
  start = Yog.Builder.GridGraph.coord_to_id(maze, 0, 0)
  exit = Yog.Builder.GridGraph.coord_to_id(maze, size - 1, size - 1)
  
  {:ok, path} = Yog.Pathfinding.shortest_path(in: g, from: start, to: exit)
  
  IO.puts("#{String.pad_trailing(name, 22)}: Shortest path = #{path.weight} steps")
end

